
Einführung ABAP-Objects

· Objektorientierte Erweiterung zu ABAP

· Klasse / Objekte können in jedem beliebigen ABAP-Programm verwendet werden

· KEINE Möglichkeit rein Objektorientiert zu arbeiten
Erstellung einer Klasse

2 Sektionen im Quelltext

· Klassendefinition

· Deklaration der Attribute

· Schnitttstellendefinition der Methoden

· Muss im Quelltext vor der ersten Verwendung der Klasse stehen

· Klassenimplementierung

· Implementierung aller definierten Methoden

· Muss im Quelltext nach allen Ausführbaren Anweisungen stehen

Klassendefinition

CLASS Klassenname DEFINITION.

PUBLIC SECTION.


Definition aller öffentlichen Attribute und Methoden (überall sichtbar)

PROTECTED SECTION.

Definition aller geschützten Attribute und Methoden (nur in abgeleiteten Klassen sichtbar)

PRIVATE SECTION.


Definition aller privaten Attribute und Methoden (nach außen nicht sichtbar)

ENDCLASS.

Klassenimplementierung

CLASS Klassenname IMPLEMENTATION.

METHOD Methodenname. 
Parameter werden NICHT angegeben, sondern so wie sie in den Klassendefinitionen deklariert werden verwendet.
Anweisungen die ausgeführt werden sollen, wenn die Methode aufgerufen wird.

ENDMETHOD.

...weitere Methodenimplementierungen

ENDCLASS.

Definition von Attributen

DATA Attributname TYPE Datentyp. 
( Objektattribut, d.h. Attribut existiert für jedes Objekt, das aus der Klasse erzeugt wird und kann für jedes Objekt unterschiedliche Werte annehmen.

CLASS-DATA Attributname TYPE Datentyp. ( 
statisches Attribut (Klassenvariable), d.h. 

Attribut existiert nur einmal für alle Objekte der Klasse und abgeleitete oder Vererbte Klassen und kann nur einen einzigen Wert annehmen

Definition von Methoden

METHODS Methodenname Parameterliste. ( Objektmethode

Methodennamen MÜSSEN innerhalb der Klasse eindeutig sein. Das konzept des Überladens von Methoden existiert NICHT.

Parameterliste:


4 Arten von Parametern:

· Importing: Übergabe von Werten, die nur gelesen werden sollen

· Exporting: Übergabe von Variablen, die beschrieben werden sollen

· Changing: Übergabe von Variablen , die gelesen und danach geändert werden sollen

· Returning: einzelner Wert, den die Methode als ergebnis zurück liefern soll

CLASS-METHODS Methodenname Parameterliste. ( Klassenmethode

Erzeugung und Verwendung von Objekten

Erzeugung von Objekten 

2 Schritte: 

1. Deklaration einer Objektvariable

DATA Variablenname TYPE REF TO Klassenname.
( Erzeugt einen Referenzvariable (Zeigervariable) die auf ein Objekt der angegebenen Klasse zeigen kann.


2. Erzeugung des Objektes

CREATE OBJEKT Variablenname Parameterliste




( Es wird ein Objekt der Klasse erzeugt und der angegebenen Variable zugewiesen

Beispiel: 
DATA auto1 TYPE REF TO Auto.



CREATE OBJECT auto1.

Verwendung des Objekts

· Zugriff auf Attribute

· Objektattribute

auto1->farbe = ‚rot’.
WRITE auto->farbe.


· Statische Attribute

Klassenname ( Attributname
KEIN Zugriff über Objektname möglich


· Aufruf von Methoden

· Objektmethoden
CALL METHOD Objektname->Methodenname Parameterliste.


· Angabe der Parametertypen

· Exporting – Parameter der in der Methode als Importing deklariert ist

· Importing – Parameter der in der Methode als Exporting
deklariert ist

· Changing und Returning wie in der Methode deklariert

CALL METHOD Objektname->Methodenname EXPORTING formaler Parameter = aktuelle Parameter.

· statische Methoden

CALL METHOD Klassenname=>Methodenname Parameterliste.

Konstruktoren

· Objektkonstruktor
Wird automatisch nach der Erzeugung eines Objektes der Klasse aufgerufen und belegt die Attribute des Objektes mit Startwerten.

CLASS auto DEFINITION.
PUBLIC SECTION.

METHODS CONSTRUCTOR IMPORTING h TYPE string t TYPE string f TYPE string.

ENDCLASS.




· statischer Konstruktor

· statische Methode

· nur zugriff auf statische Elemente

· wird automtisch aufgerufen: vor dem ersten Zugriff auf ein Element der Klasse (auch wenn noch kein Objekt erzeugt wurde)

CLASS auto

PUBLIC SECTION.

...

CLASS-METHODS CLASS_CONSTRUCTOR. 

…

ENDCLASS.

CLASS auto IMPLEMENTATION.

METHOD CLASS_CONSTRUCTOR.


WRITE ‘Aufruf Klassenkonstruktor’.

ENDMETHOD.

ENDCLASS.

Vererbung in ABAP

Eine Klasse kann die Eigenschaften und Methoden einer anderen Klasse erben.

· Es existiert nur Einfachvererbung, d.h. eine abgeleitete Klasse kann nur von einer einzigen Basisklasse erben

· Jede Klasse für die KEINE  Vererbung definiert wird erbt automatisch von der vordefinierten Klasse OBJECT

· Es werden immer alle Komponenten einer Basisklasse vererbt

Syntax

CLASS Klassenname DEFINITION INHERITING FROM Basisklassenname.

Es wird eine neue Klasse (abgebildete Klasse) erstellt, die alle Attribute und Methoden der Basisklasse erbt.

Vorwärtsdeklaration

REPORT  Z80304_VORWAERTSDEKL                                        .

CLASS k2 DEFINITION DEFERRED.

class k1 definition.

  public section.

    data va type ref to k2. "#
endclass.

*        # Bezug auf Klasse 2 k2, aber Klasse k2 wird erst später   

*          definiert

*Problem: Vorwärtsdeklaration in ABAP

*         =>jedes Element MUSS vor seiner Verwendung definiert werden

*Lösung: Klasse MUSS vor Ihrer Verwendung bekannt gemacht werden

*

*        CLASS Klassenname DEFINITION DEFERRED.

*

*Ausgabe, dass die Klasseendefinition auf einen späteren Zeitpunkt *verschoben wird.

*=>Es können Variablen von Typ der Klasse erzeugt werden        

*vor der entgültigen Definition

* - können KEINE Objekte der Klasse erzeugt werden

* - kann NICHT auf statische Komponenten zugegriffen werden
class k2 definition.

  public section.

    data v2 type ref to k1.

endclass.

class k1 implementation.

endclass.

class k2 implementation.

endclass.

Abstrakte Klassen

1.Fall: 
Eine Klasse soll NUR zur Vererbung verwendet werden, d.h. es soll keine Möglichkeit 

bestehen, aus diesen Klassen Objekte zu erzeugen

2.Fall:
In einer Klasse sollen die Schnittstellen für Methoden definiert werden, aber die 

Implementierungen sollen in einer abgeleiteten Klasse realisiert werden.

Definition einer abstrakten Klasse


CLASS Klassenname DEFENITION ABSTRACT.



...


ENDCLASS.

Beispiel zu Fall 1

class k1 definition abstract.

public section.

  data a type i value 5.

  methods m1.

endclass.

class k2 definition inheriting from k1.

public section.

  data b type i value 10.

  methods m2.

endclass.

*__________________________________

data obj1 type ref to k1.

*Es kann eine Variable angelegt werden, die auf ein Objekt einer *abstrakten Klasse zeigen kann

data obj2 type ref to k2.

create object obj1.

*FEHLER: Es kann kein Object einer abstrakten Klasse erzeugt werden

create object obj2.

call method obj2->m1.

*geerbte Methode aus k1

call method obj2->m2.

*implementiert in k2

*call method obj1->m1.

*Laufzeitfehler: Null Pointer

*__________________________________

class k1 implementation.

method m1.

  write: 'a= ',a.

endmethod.

endclass.

class k2 implementation.

method m2.

  write: 'a= ',a,' b= ',b.

endmethod.

endclass.

Beispiel zu Fall 2

class k1 definition abstract.

*Ein Klasse die mindestens eine abstrakte Methode enthällt, MUSS mit dem

*Schlüsselwort „abstract“ definiert werden

Public section.

  Data a type I value 5.

  Methods: m1,

           m2 abstract.

*Eine Methode die mit abstract gekennzeichnet ist, darf nicht im

*Implementationsteil der Klasse implementiert werden, sondern die

*Implementierung wird auf eine abgeleitete Klasse verlagert

Endclass.

class k2 definition inheriting from k1.

*Eine Klasse die von einer abstrakten Klasse erbt, muss entweder alle

*geerbten abstrakten Methoden implementieren oder selbst als abstrakt

*gekennzeichnet werden

Public section.

  Data b type I value 10.

  Methods:m3,

          m2 redefinition.

*         Eine geerbte abstrakte Methode die implementiert werden soll,

*         muss in der Klassendefinition aufgeführt und mit „redefinition“

*         gekennzeichnet werden

endclass.

*___________________________________

data obj1 type ref to k1.

data obj2 type ref to k2.

*create object obj1. FEHLER: k1 ist abstrakt

*create object obj2.

call method obj2->m1.

call method obj2->m2.

call method obj2->m3.

*call method obj1->m1. beide syntaktisch korrekt, aber in

*call method obj1->m2. beiden Fällen Laufzeitfehler, obj1 zeigt auf null

*___________________________________

class k1 implementation.

*Implementierung kann weggelassen werden wenn alle Methoden abstrakt

*sind

method m1.

  write: 'm1 aus k1'.

*m2 darf nicht implementiert werden, da die methode abstract ist

endmethod.

endclass.

class k2 implementation.

  method m2.

    write 'm2 definiert in k1 und implementiert in k2'.

  endmethod.

  method m3.

    write 'm3 aus k2'.

  endmethod.

endclass.

Beispiel 3

class k1 definition abstract.

public section.

  methods: m1 abstract,

           m2 abstract importing a type i .

endclass.

class k2 definition inheriting from k1 abstract.

*Klasse muss abstrakt sein, da nicht geerbten abstrakten Methoden 

*implementiert werden

public section.

  methods: m1 redefinition,

           m3.

endclass.

class k3 definition inheriting from k2.

*alle geerbten abstrakten Methoden sind implementiert 

*(=> muss NICHT abstrakt sein)

public section.

  methods: m2 redefinition.

*          Parameter werden nicht angegeben, sondern aus ursprünglichen 

*          definition übernommen           

endclass.

*______________________________________________

data obj type ref to k3.

create object obj.

call method obj->m1.

call method obj->m2 exporting a = 5.

call method obj->m3.

*______________________________________________

class k2 implementation.

  method m1.

    write 'm1'.

  endmethod.

  method m3.

    write 'm3'.

  endmethod.

endclass.

class k3 implementation.

  method m2.

    write: 'm2 ',a.

  endmethod.

endclass.

Überschreiben von Methoden

Überschreiben bedeutet dass in einer abgeleiteten Klasse eine geerbte Methode implementiert wird.

Regeln für das überschreiben:

· Methodenname muss identisch sein

· Signatur der Methode darf sich nicht ändern

· Das Konzept des Überladen von Methoden existiert in ABAP-Objects nicht

· Sichtbarkeit der Methode darf nicht verändert werden

Beim Aufruf einer überschriebenen Methode wird immer die Methode der Klasse aufgerufen, aus der das Objekt erzeugt wurde.

REPORT  Z80304_UEBERSCHREIB                                         .

class k1 definition.

public section.

  methods m1 importing a type i b type i.

endclass.

class k2 definition inheriting from k1.

public section.

  methods m1 redefinition.

*  geerbte Methode wird neu implementiert

* (Parameter dürfen nicht angegeben werden)

endclass.

*************************************************

data: obj1 type ref to k1,

      obj2 type ref to k2.

create object obj1.

create object obj2.

call method obj1->m1 exporting a = 4 b = 1.

call method obj2->m1 exporting a = 4 b = 1.

call method obj1->m1 exporting a = 1 b = 4.

call method obj2->m1 exporting a = 1 b = 4.

*AUSGABE: 3, 3, 3-, 0

*************************************************

class k1 implementation.

  method m1.

    data erg type i.

    erg = a - b.

    write: 'k1 erg= ', erg.

  endmethod.

endclass.

class k2 implementation.

  method m1.

    data erg type i.

    erg = a - b.

    if erg < 0.

      erg = 0.

    endif.

    write: 'k2 erg= ', erg.

  endmethod.

endclass.

Zugriff auf die überschriebene Methode der Basisklasse

REPORT  Z80304_UEBERSCHREIB2                                        .

class k1 definition.

  public section.

    methods m1.

endclass.

class k2 definition inheriting from k1.

  public section.

    methods: m1 redefinition,

             m2.

endclass.

**********************************************

data: obj1 type ref to k1,

      obj2 type ref to k2.

create object obj1.

create object obj2.

call method obj1->m1.

call method obj2->m1.

call method obj2->m2.

*AUSGABE: m1 aus k1, m1 aus k2, m1 aus k1 (durch super-Aufruf),

*         m1 aus k2 (durch Aufruf von m1 innerhalb von m2),

*         m1 aus k1 (weil innerhalb von m1 erneut über super m1 aus k1 

*                    aufgerufen wird)

**********************************************

class k1 implementation.

  method m1.

    write 'm1 aus k1'.

  endmethod.

endclass.

class k2 implementation.

  method m1.

    write 'm1 aus k2'.

    call method super->m1.

*Aufruf der überschriebenen Methode aus k1.

*überschriebene Methode m1 ist nur aus Methode aufrufbar, welche sie

*überschrieben hat

  endmethod.

  method m2.

    write 'm2 aus k2'.

    call method m1.

*Aufruf der neu implementierten Methode aus k2

  endmethod.

Verhindern von Vererbung bzw. Überschreiben

1. Verhindern von Vererbung:

REPORT  Z80304_UEBERSCHREIB2VERH                                    .

class k1 definition final.

*FINAL schliesst eine Vererbungshierarchie ab, d.h. von einer Klasse die *mit FINAL deklariert wurde, kann KEINE abgeleitete Klasse erstellt *werden

public section.

  data a type i.

endclass.

class k2 definition inheriting from k1.

public section.

  data b type i.

endclass.

2. Verhindern von Überschreiben

class k3 definition.

public section.

  Methods: m1,

           m2 final,

           m3 final importing a type i.

endklss.

class k4 definition inhriting from k3.

public section.

  methods: m1 redefinition,

           m2 redefinition,

           m3 redefinition.

* Methoden m2 und m3 sind final und können nicht überschrieben werden.

endclass.

Regeln für den Einsatz von final bei Methodendefinitionen:

· In Klassen die mit Final gekennzeichnet sind, darf das Schlüsselwort final nicht für Methoden verwendet werden (Methoden sind automatisch final)

· Kombination von final und abstrakt ist ungültig

· Konstruktoren von Klassen sind automatisch final, das Schlüsselwort final ist optional

· Private Methoden sollten (SAP) nicht mit final gekennzeichnet werden, da sie in einer abgeleiteten Klasse nicht sichtbar sind

optional





optional





Name einer vorher definierten Objektvariable





Prarametername in der Methodendeklaration





Wert der übergeben werden soll





Liste aller Parameter, die bei der Erzeugung eines Objektes angegeben werden sollen.








